

Chapitre 1

Initiation à Ocaml

Ocaml est un langage de programmation relevant notamment du paradigme de la programmation fonctionnelle, c'est à dire qu'il repose sur l'évaluation de fonctions, mais permettant également d'adopter un style de programmation impérative.

Un programme Ocaml est une suite d'expressions et de déclarations séparées par un double point-virgule ; ; dans lesquelles l'indentation et les retours à la ligne n'a pas d'importance mais facilitent grandement la lisibilité et la compréhension du code.

Une expression peut elle-même être une séquence de plusieurs expressions séparées d'un point-virgule ; . Dans ce cas c'est l'évaluation de la dernière expression qui est renvoyée.

1.1 Variables

1.1.1 Déclaration et types

La déclaration d'une variable s'effectue en Ocaml grâce au mot-clé `let` selon la syntaxe suivante :

```
let ma_variable = ma_valeur;;
```

Exemple 1.1.

```
# let a = 8;;  
val a : int = 8
```

La sortie indique qu'on vient de créer une *variable* dont :

- le *nom* est a
- le *type* est int
- la *valeur* est 8

Pour faciliter la compréhension du code on choisit des noms de variables explicites lorsque cela fait sens (comme `prix` ou `distance` plutôt que `a` ou `x`).

Dans tous les cas on choisira des mots *sans espaces* et *non accentués* (on pourra utiliser le symbole `_`).

Exemple 1.2.

```
# let duree_en_h = 3.5;;  
val duree_en_h : float = 3.5
```

Une variable définie comme ci-dessus est *immuable* (sa valeur n'est pas modifiable) et possède obligatoirement un *type*, on parle de *typage statique*.

Le type est *inféré* par le compilateur ou l'interpréteur : il est déduit des différentes fonctions et opérateurs impliqués dans l'expression évaluée lors de la déclaration.

Le tableau suivant récapitule les types de base et les opérateurs associés :

	Type	Opérateur	
Entiers relatifs	int	+	addition
		-	soustraction
		*	multiplication
		/	division euclidienne
		mod	reste de la division euclidienne (si positifs)
Nombres à virgule flottante	float	+.	addition
		-.	soustraction
		*.	multiplication
		/.	division exacte
		**	puissance
Booléen (true ou false)	bool	not	négation
		&&	ET logique
			OU logique
Caractère	char		ex: 'c'
Chaîne de caractères	string	^	concaténation : "une "^"chaîne" = "une chaîne"
Rien	unit		() est de type unit

Remarque 1.1. Les nombres à virgule flottante (ou *flottants*) s'écrivent avec un point et non une virgule ! Sinon, il s'agira d'un couple d'entiers (type `int * int`).

Exemple 1.3.

```
# 2.718;;
- : float = 2.718
# 2,718;;
- : int * int = (2, 718)
```

Remarque 1.2. Pour les caractères on utilise des guillemets simples (') tandis que pour les chaînes de caractères on utilise des guillemets doubles (").

Remarque 1.3. La longueur d'une chaîne de caractères s'obtient avec `String.length`. L'accès aux différents caractères de la chaîne se fait selon la syntaxe `s.[i]` où `i` est l'indice souhaité (à partir de l'indice 0), mais on ne peut pas les modifier : on dit que les chaînes de caractères sont *immuables*.

Si besoin, on pourra convertir une expression en son équivalent d'un autre type. On dispose notamment des fonctions de conversion suivantes :

Type de départ	Type d'arrivée	Fonction	Exemple	Correspondance
int	float	float_of_int	4	4.
float	int	int_of_float (troncature à l'unité)	3.14	3
int	string	string_of_int	4	"4"
float	string	string_of_float	3.14	"3.14"
int	char	char_of_int (correspondance ASCII)	64	'@'
char	int	int_of_char (correspondance ASCII)	'a'	97

1.1.2 Overflow

Les entier et les flottants sont sujets au dépassement de capacité. En effet, les variables sont stockées dans la mémoire RAM de l'ordinateur, or celle-ci n'est pas infinie.

- Le plus grand entier représentable est stocké dans la variable `max_int`, si on augmente ce nombre de 1, on tombe sur le plus petit entier représentable.

On parle de dépassement d'entier ou encore d'*integer overflow*.

- De même, `max_float` est le plus grand flottant représentable.

Exemple 1.4.

```
# max_int;;
- : int = 4611686018427387903
# max_int + 1;;
- : int = -4611686018427387904
# max_float;;
- : float = 1.79769313486231571e+308
```

Remarque 1.4. Le dépassement d'entier est notamment responsable du crash de la fusée Ariane 5...

1.1.3 Comparaisons et type bool

Pour effectuer une comparaison entre deux objets de même type on utilise les opérateurs =, <, >, <=, >= ou <> (pour tester la différence).

Le résultat est de type bool : true (vrai) ou false (faux).

Exemple 1.5.

```
# 4.2 < 2.4;;
- : bool = false
# 7 >= (-2);;
- : bool = true
# 0.1 +. 0.2 <> 0.3;;
- : bool = true (* ce resultat est plus etonnant, mais deja vu en TP *)
# 64 mod 5 = 0;;
- : bool = false
```

Les opérateurs && et || effectuent une *évaluation paresseuse* des expressions situées de part et d'autre, cela signifie que :

- pour évaluer l'expression a && b (ET logique), c'est l'expression a qui est évaluée la première.

→ si le résultat est true, l'évaluation de b est retournée
→ sinon c'est directement false qui est retourné

- pour évaluer l'expression a || b (OU logique), c'est l'expression a qui est évaluée la première.

→ si le résultat est false, l'évaluation de b est retournée
→ sinon c'est directement true qui est retourné

Exemple 1.6.

```
# 4 < 6 && 1024 <= 1024;;
- : bool = true
# 5.2 > 8. || 13 mod 2 = 0;;
- : bool = false
# not (4 = 7);;
- : bool = true
```

1.1.4 Variables locales

La construction vue ci-dessus permet de définir une *variable globale*.

Une variable est dite *locale* lorsqu'elle n'est définie que dans le cadre d'une expression.

On utilise pour cela la construction let ... in

Exemple 1.7. On souhaite convertir 47 heures en minutes :

```
# let min_par_h = 60 in
    47 * min_par_h;;
- : int = 2820
# min_par_h;;
Error: Unbound value min_par_h      (* min_par_h n'est plus définie *)
```

1.1.5 Références

On a vu que les variables définies comme précédemment étaient *immuables* (leur valeur n'est pas modifiable).

Lorsqu'on souhaite utiliser des variables *mutables* on utilise des *références*, que l'on déclare à l'aide du mot-clé `ref` en adoptant la syntaxe suivante :

```
let ma_ref = ref ma_valeur;;
```

- L'accès à la valeur de la référence se fait en ajoutant un `!` devant son nom : `!ma_ref`
- L'assignation d'une nouvelle valeur s'effectue selon la syntaxe `ma_ref := autre_valeur;;`

Exemple 1.8.

```
# let x = ref 3;;
val x : int ref = {contents = 3}
# x := 4 * !x;;
- : unit = ()
# x;;
- : int ref = {contents = 12}
# !x;;
- : int = 12
```

Remarque 1.5. Une référence dont la valeur est de type `int` est de type `int ref`.

Pour bien comprendre la différence entre une variable standard et une référence, voici un exemple d'évaluation d'expressions a priori équivalentes :

Sans référence :

```
# let x = 4;;
val x : int = 4
# let add y = x + y;;
val add : int -> int = <fun>
# add 8;;
- : int = 12
# let x = 32;;
val x : int = 32
# add 8;;
- : int = 12
```

Avec référence :

```
# let x = ref 4;;
val x : int ref = {contents = 4}
# let add y = !x + y;;
val add : int -> int = <fun>
# add 8;;
- : int = 12
# x := 32;;
- : unit = ()
# add 8;;
- : int = 40
```

Remarque 1.6. On n'utilisera des références que lorsque c'est vraiment nécessaire.

1.2 Fonctions

1.2.1 Déclaration et signature

On rappelle que Ocaml est avant tout un langage de *programmation fonctionnelle*. Les fonctions y occupent donc naturellement une place importante.

Une fonction Ocaml ne devrait pas générer d'**effet de bord**. On parle d'effet de bord lorsque l'évaluation d'une expression (dite **impure**) affecte l'environnement extérieur à cette expression, par exemple en modifiant la valeur d'une variable globale, en faisant de l'affichage sur écran ou bien en modifiant le contenu d'un fichier.

Pour déclarer une fonction on utilise la syntaxe suivante : `let fonction argument = expression;;`

Exemple 1.9.

```
# let carre x = x * x;;
val carre : int -> int = <fun>
# carre 12;;          (* inutile de mettre des parentheses *)
- : int = 144
# carre (-9);;        (* sauf cas particuliers *)
- : int = 81
```

Une fonction possède une **signature**, inférée par le compilateur ou l'interpréteur.

En notant `type_entree` le type de argument et `type_sortie` le type de expression, la signature de la fonction est : `type_entree -> type_sortie`

Exemple 1.10. La fonction `carre` définie plus haut a pour signature `int -> int`.

Remarque 1.7. Lorsqu'on évalue l'expression `carre (3 * 4)` l'expression `3 * 4` est d'abord évaluée puis le résultat est passé comme argument à la fonction. C'est donc `carre 12` qui est évalué.

On dit que Ocaml est un **langage strict** : l'évaluation des arguments se font avant l'appel de la fonction, contrairement aux langages dits paresseux pour lesquels l'évaluation d'un argument n'est effectuée qu'au moment de sa première utilisation. (Voir le fonctionnement des opérateurs `&&` et `||` page 7)

Une fonction peut tout à fait ne pas avoir d'argument.

Exemple 1.11.

La fonction `reset_x` déclarée ci-dessous, de type `unit`, permet de réinitialiser la référence `x` à la valeur 0.

```
# let x = ref 12;;
val x : int ref = {contents = 12}
# let reset_x () = x := 0 ;;
val reset_x : unit -> unit = <fun>
# reset_x ();;
- : unit = ()
# x;;
- : int ref = {contents = 0}
```

Exemple 1.12. La fonction `swap` déclarée ci-dessous, de type `unit`, permet d'échanger les valeurs de deux références.

```
# let swap x y = let z = !x in      (* la variable locale z stocke la valeur de x *)
  x := !y;
  y := z
;;
val swap : 'a ref -> 'a ref -> unit = <fun>
# let x, y = ref 16, ref 64;;
val x : int ref = {contents = 16}
val y : int ref = {contents = 64}
# swap x y;;
- : unit = ()
# x, y;;
- : int ref * int ref = ({contents = 64}, {contents = 16})
```

On notera que l'expression `x, y;;` est de type `int ref * int ref` : il s'agit d'un couple de références sur entiers.

Pour déclarer une fonction à plusieurs variables on utilise la syntaxe :

```
let fonction arg1 arg2 arg3 ... = expression;;
```

La signature d'une telle fonction est alors de la forme : `type1 -> type2 -> type -> ... -> type_sortie`

En réalité il s'agit d'une suite de fonctions à un seul argument imbriquées entre elles. C'est ce qu'on appelle la **curryfication**. Un avantage notable est qu'on peut appliquer partiellement une fonction.

Exemple 1.13.

```
# let prod x y = x * y;;
val prod : int -> int -> int = <fun>
# prod 2;;
- : int -> int = <fun>
# let double = prod 2;;
val double : int -> int = <fun>
# double 3;;
- : int = 6
```

On trouve fréquemment des variables locales dans des fonctions.

Exemple 1.14.

```
# let conversion_h_min heure = let min_par_h = 60 in
    heure * min_par_h;;
val conversion_h_min : int -> int = <fun>
# conversion_h_min 3;;
- : int = 180
```

1.2.2 Polymorphisme

Il peut arriver que les types ne puissent pas être inférés car il n'y a pas de contrainte particulière induite par l'expression évaluée (contrairement à l'utilisation de l'opérateur `+` par exemple).

Dans ce cas on note les différents types possibles `'a`, `'b`, etc. On conserve la même notation lorsque deux types doivent être identiques dans une signature.

Exemple 1.15.

```
# let first x y = x;;
val first : 'a -> 'b -> 'a = <fun>
# let second x y = y;;
val second : 'a -> 'b -> 'b = <fun>
# first 4 8;;
- : int = 4
# second true false;;
- : bool = false
```

1.2.3 Type unit

Certaines expressions ne renvoient aucune valeur après évaluation, ce qui se note `()`. Ces expressions sont de type `unit`. Celles-ci peuvent en revanche générer des effets de bord. C'est le cas par exemple lorsqu'une expression utilise l'une des fonctions `print_int`, `print_float`, `print_string` ou `print_newline` qui permettent de faire de l'affichage sur écran.

Exemple 1.16.

```
# let a = 4;;
val a : int = 4
# print_string "la racine de ";
  print_int (a*a);
  print_string " vaut ";
  print_int a;
  print_newline ();;
la racine de 16 vaut 4
- : unit = ()
```

Une fonction dont le résultat est de type `unit` est appelée une **procédure**. Les fonctions `reset_x` et `swap` des exemples 1.11 et 1.12 (page 9) sont également des procédures. L'appel de ces fonctions n'engendre pas d'affichage mais modifie l'environnement extérieur en modifiant les valeurs de références.

Les fonctions `read_int`, `read_float` et `read_line`, permettent quant à elles de lire respectivement un entier, un flottant et une chaîne de caractères saisie par l'utilisateur.

Exemple 1.17.

```
# let nb = read_int () in
  print_string "Le double de ";
  print_int nb;
  print_string " vaut ";
  print_int (2 * nb);
  print_newline ();;
16          (* cette valeur est saisie par l'utilisateur *)
Le double de 16 vaut 32
- : unit = ()
```

1.2.4 Fonctions anonymes - ordre supérieur

Il n'est pas toujours utile de donner un nom à une fonction. C'est le cas par exemple si l'on souhaite n'appeler cette fonction qu'une seule fois. On utilise alors la syntaxe suivante : `fun x -> ...`

Exemple 1.18.

```
# (fun x -> x * x - x) 8;;
- : int = 56
```

Ainsi, il est tout à fait équivalent d'écrire : `let f x = expr` et `let f = fun x -> expr`

En fait, une fonction Ocaml est une valeur comme les autres. La deuxième syntaxe ci-dessus est d'ailleurs tout à fait identique à celle utilisée pour déclarer des variables de l'un des types vus précédemment. On peut donc tout à fait passer une fonction comme argument d'une autre fonction.

On dit qu'une fonction Ocaml est d'**ordre supérieur** lorsqu'elle prend en argument une ou plusieurs fonctions.

Exemple 1.19.

```
# let eval f g = (f 0) * (g 0);;
val eval : (int -> int) -> (int -> int) -> int = <fun>
# eval (fun x -> x + 2) (fun x -> 2 * x - 4);;
- : int = -8
```

1.2.5 Fonctions récursives

Une fonction est dite **récursive** lorsqu'elle est déclarée en utilisant une expression faisant appel à elle-même.

On déclare une fonction récursive Ocaml à l'aide du mot-clé `rec` :

```
let rec ma_fonction arg = ...
```

Exemple 1.20. La fonction ci-dessous permet de calculer la factorielle d'un nombre :

$$!n = \begin{cases} 1 & \text{si } n = 0 \\ n \times !(n-1) & \text{sinon} \end{cases}$$

```
# let rec factorielle n =  
    if n = 0 then 1  
    else n * factorielle (n-1);;  
val factorielle : int -> int = <fun>  
# factorielle 5;;  
- : int = 120
```

Remarque 1.8. Attention, lorsqu'on écrit une fonction récursive il faut s'assurer d'avoir un « cas de base » afin d'être certain que l'appel de cette fonction ne générera pas une infinité d'appels à cette même fonction. Dans le cas de la fonction `factorielle` définie ci-dessus, il s'agit du cas $n = 0$.

Remarque 1.9. Une même fonction mathématique peut être implémentée de différentes façons, de manière récursive ou non. Nous verrons dans un prochain chapitre qu'il faudra rester vigilant notamment quant à la complexité en temps. En effet, une fonction récursive peut générer un arbre d'appels important rendant l'exécution impossible en un temps acceptable.

1.3 Structures de contrôle

1.3.1 Structure conditionnelle `if ... then ... else ...`

L'expression `if condition then expr1 else expr2` évalue d'abord l'expression `condition`, qui doit être un booléen. Si le résultat est `true`, c'est `expr1` qui est évaluée puis retournée. Sinon, c'est `expr2`.

Exemple 1.21.

```
# let abs x = if x > 0.  
              then x  
              else -. x  
;;  
val abs : float -> float = <fun>
```

Remarque 1.10. Le `else` n'est pas toujours nécessaire. On n'écrit pas de `else` s'il n'y a rien à évaluer dans le cas où la condition ne serait pas satisfaite.

Remarque 1.11. Si on souhaite distinguer plus de deux cas on peut imbriquer d'autres `if` :

```
if condition1 then expr1 else  
if condition2 then expr2 else  
...  
if conditionp then exprp else expression;;
```


1.3.2 begin ... end

Lorsqu'une expression, suivant par exemple un then, est une *séquence* de plusieurs expressions, il est nécessaire d'utiliser des parenthèses ou bien la structure begin ... end afin de grouper ces expressions.

Ainsi, on écrira au choix :

```
if condition then
    (expr1;
     expr2;
     expr3)
```

ou bien

```
if condition then
    begin
        expr1;
        expr2;
        expr3
    end
```

À défaut, malgré une bonne indentation (qui on le rappelle n'a aucune importance), cela sera interprété comme :

```
(if condition then
    expr1);
    expr2;
    expr3
```

1.3.3 Structure itérative : boucle for

Une boucle for a la syntaxe suivante :

```
for var = indice_debut to indice_fin do
    expression
done
```

Exemple 1.22.

```
# for i = 1 to 9 do
    print_string "Le carre de ";
    print_int i;
    print_string " vaut ";
    print_int (i*i);
    print_newline ()
done;;
Le carre de 1 vaut 1
Le carre de 2 vaut 4
Le carre de 3 vaut 9
Le carre de 4 vaut 16
Le carre de 5 vaut 25
Le carre de 6 vaut 36
Le carre de 7 vaut 49
Le carre de 8 vaut 64
Le carre de 9 vaut 81
- : unit = ()
```

1.3.4 Structure itérative : boucle while

Une boucle while a la syntaxe suivante :

```
while condition do
  expression
done
```

Exemple 1.23.

```
# let pgcd n m = let q, r = ref n, ref m in
  while (!r <> 0) do
    let reste = !q mod !r in
    q := !r;
    r := reste
  done;
  !q;;
val euclide : int -> int -> int = <fun>
# pgcd 256 48;;
- : int = 16
# pgcd 246 13;;
- : int = 1
```

On peut utiliser la fonction pgcd pour déclarer une fonction premiers_entre_eux qui détermine si deux entiers naturels sont premiers entre eux ou non :

```
# let premiers_entre_eux n m = (pgcd n m = 1);;
val premiers_entre_eux : int -> int -> bool = <fun>
# premiers_entre_eux 256 48;;
- : bool = false
# premiers_entre_eux 246 13;;
- : bool = true
```

1.4 Types structurés prédéfinis

1.4.1 n-uplets

Un n -uplet Ocaml, est formé de n valeurs, de type éventuellement différents, séparées par des virgules. On déclare un n -uplet selon la syntaxe suivante :

```
let mon_n_uplet = val1, ..., valn;;
```

On obtient alors une variable de type `type1 * ... * typen` où `type1`, ..., `typen` sont les types des valeurs formant le n -uplet.

Exemple 1.24.

```
# let tuple = (5, true, "hello", '@');;
val tuple : int * bool * string * char = (5, true, "hello", '@')
```

Pour accéder aux différents éléments d'un n -uplet, il faut procéder à une déclaration **déstructurante**.

Exemple 1.25.

```
# let (val1, val2, val3, val4) = tuple;;
val val1 : int = 5
val val2 : bool = true
val val3 : string = "hello"
val val4 : char = '@'
```

Remarque 1.12. Les éléments d'un n -uplet ne sont pas modifiables en place, ils sont **immuables**.

1.4.2 Listes

Une liste Ocaml est construite à partir de la liste vide `[]`, à laquelle on a éventuellement ajouté des éléments en tête de liste, tous de même type, avec l'opérateur `::` (prononcer « cons »).

On obtient alors une liste de type `'a list` où `'a` correspond au type de tous les éléments de la liste.

Une liste de n éléments peut ainsi être déclarée selon la syntaxe suivante :

```
let ma_liste = e1 :: e2 :: ... :: en :: [];;
```

Une autre syntaxe possible est la suivante : `let ma_liste = [e1; e2; ...; en];;`

La fonction `List.length` permet d'accéder à la longueur d'une liste.

Exemple 1.26.

```
# let premiers = [2; 3; 5; 7; 11; 13; 17; 19];;
val premiers : int list = [2; 3; 5; 7; 11; 13; 17; 19]
# let voyelles = 'a' :: 'e' :: 'i' :: 'o' :: 'u' :: 'y' :: [];;
val voyelles : char list = ['a'; 'e'; 'i'; 'o'; 'u'; 'y']
# List.length voyelles;;
- : int = 6
```

Les liste Ocaml sont *immuables*. L'opérateur `::` (« cons ») est utilisé pour construire une nouvelle liste comportant un élément supplémentaire en tête de liste, mais ne permet pas de modifier une liste.

Exemple 1.27.

```
# let liste = ["farine"; "oeufs"; "lait"; "huile"];;
val liste : string list = ["farine"; "oeufs"; "lait"; "huile"]
# "chocolat"::liste;;
- : string list = ["chocolat"; "farine"; "oeufs"; "lait"; "huile"]
# liste;;
- : string list = ["farine"; "oeufs"; "lait"; "huile"]
```

Contrairement aux n -uplets ou aux chaînes de caractères, on ne peut pas accéder à un élément quelconque d'une liste sans passer par une fonction récursive (voir section ?? page ??).

Enfin, deux listes peuvent être concaténées en une nouvelle liste grâce à l'opérateur `@` mais attention, la complexité est en $\mathcal{O}(n)$ où n est la longueur de la seconde liste (voir prochain chapitre).

Exemple 1.28.

```
# [1.; 0.1; 0.01; 0.001] @ [0.0001; 0.00001];;
- : float list = [1.; 0.1; 0.01; 0.001; 0.0001; 1e-05]
```

1.4.3 Tableaux

Un tableau Ocaml est constitué d'un nombre fixé d'éléments, tous de même type.

Un tableau est de type `'a array` où `'a` correspond au type de tous les éléments du tableau.

Un tableau de n éléments peut être déclaré selon la syntaxe suivante :

```
let mon_tableau = [| e1; e2; ...; en |];;
```

Remarque 1.13. Le tableau vide se note `[| |]` .

La fonction `Array.length` permet d'accéder à la longueur d'un tableau.

Exemple 1.29.

```
# let tab = [| 8; 15; 6; 14; 18|];;
val tab : int array = [|8; 15; 6; 14; 18|]
# Array.length tab;;
- : int = 5
```

Contrairement aux listes, un tableau est *mutable* : on peut modifier en place ses éléments.

- `tab.(i)` permet d'accéder à l'élément `i` du tableau `tab` (à partir de l'indice 0).
- `tab.(i) <- new` permet de remplacer l'élément d'indice `i` par la valeur `new`.

On peut également créer un tableau en fixant son nombre d'éléments et en les initialisant tous à la même valeur.

Exemple 1.30.

```
# let etat = Array.make 8 true;;
val etat : bool array = [|true; true; true; true; true; true; true; true|]
# etat.(1) <- false; etat.(4) <- false;;
- : unit = ()
# etat;;
- : bool array = [|true; false; true; true; false; true; true; true|]
# etat.(8) <- false;;
Exception: Invalid_argument "index out of bounds".
```

Remarque 1.14. Dans l'exemple ci-dessus on observe une erreur « index out of bounds » lorsqu'on cherche à accéder à l'élément d'indice 8 car les indices vont de 0 à 7 pour un tableau à 8 éléments.

Nous parlerons des exceptions dans une prochaine section.

Enfin, on peut copier un tableau grâce à la fonction `Array.copy`. Attention en revanche, il s'agit d'une copie superficielle : si on modifie les éléments du tableau initial, cela n'affecte pas la copie. En effet, les deux tableaux pointent vers les mêmes valeurs, mais occupent des espaces mémoire différents. On pourra s'en convaincre avec l'exemple ci-dessous :

Exemple 1.31.

```
# let init = [|1; 2; 3; 4|];;
val init : int array = [|1; 2; 3; 4|]
# let copie = Array.copy init;; (* copie superficielle *)
val copie : int array = [|1; 2; 3; 4|]
# copie = init;;
- : bool = true
# copie == init;; (* l'opérateur == compare les adresses memoire *)
- : bool = false
# let tab = init;;
val tab : int array = [|1; 2; 3; 4|]
# tab.(0) <- 64;;
- : unit = ()
# init.(0);; (* toute modification de tab modifie également init *)
- : int = 64
```

1.4.4 Types option

Lorsqu'on souhaite indiquer la présence ou non d'une valeur on utilise le type `'a option`. Les variables de ce type peuvent alors valoir :

- `None` : ce qui correspond à l'absence de valeur
- `Some val` : ce qui correspond à la présence de la valeur `val`

Exemple 1.32. La fonction suivante détermine si un caractère est présent dans une chaîne de caractères ou non et donne sa dernière position le cas échéant :

```
# let position chaine c =  
  let pos = ref None in  
  let long = String.length chaine in  
  for i = 0 to long - 1 do  
    if chaine.[i] = c then pos := Some i  
  done;  
!pos;;  
val position : string -> char -> int option = <fun>  
# position "une chaine" 'h';;  
- : int option = Some 5  
# position "une chaine" 'z';;  
- : int option = None
```

1.5 Filtrage de motifs

1.5.1 Syntaxe

Lorsqu'on souhaite renvoyer un résultat différent selon la valeur obtenue après l'évaluation d'une expression, on effectue un *filtrage de motifs* selon la syntaxe suivante :

```
match expression with  
  | motif1 -> resultat1  
  | motif2 -> resultat2  
  ...  
  | _ -> autre_resultat
```

Remarque 1.15. Le *motif universel* `_` signifie « tout autre motif ».

Exemple 1.33. La fonction factorielle de l'exemple 1.20 page 12 aurait pu être déclarée de la façon suivante :

```
# let rec factorielle n = match n with  
  | 0 -> 1  
  | _ -> n * (factorielle (n-1) )  
;;  
val factorielle : int -> int = <fun>  
# factorielle 5;;  
- : int = 120
```

Le mot-clé `function` permet même d'obtenir une syntaxe encore plus concise en effectuant un filtrage sur le dernier argument de la fonction.

Exemple 1.34. Toujours pour la fonction factorielle, cela donne :

```
# let rec factorielle = function  
  | 0 -> 1  
  | n -> n * (factorielle (n-1) )  
;;  
val factorielle : int -> int = <fun>  
# factorielle 5;;  
- : int = 120
```

Remarque 1.16. La correspondance de motif est recherchée dans l'ordre des motifs proposés, il est donc important de les ordonner correctement. Dans l'exemple précédent, il est impératif d'indiquer le motif « 0 » avant le cas général.

Exemple 1.35.

```
# let fruit = function
  | "poire" -> true
  | "fraise" -> true
  | "pomme" -> true
  | "courgette" -> false
  | "potimarron" -> false
;;
Warning 8: this pattern-matching is not exhaustive.
Here is an example of a value that is not matched:
"""
val fruit : string -> bool = <fun>
```

Attention à bien s'assurer que le filtrage est *exhaustif*. Ce n'est pas le cas dans cet exemple (d'où le message d'alerte). La fonction est bien déclarée mais on obtiendra par exemple :

```
# fruit "poire";;
- : bool = true
# fruit "courgette";;
- : bool = false
# fruit "cerise";;
Exception: Match_failure ("//toplevel//", 76, -26).
```

1.5.2 Le cas des listes

Le filtrage de motifs permet de parcourir des listes en utilisant des fonction récursives.

Exemple 1.36. La fonction suivante permet de compter le nombre d'occurrences d'une valeur dans une liste :

```
# let rec compte liste elem = match liste with
  | [] -> 0
  | e::q -> compte q elem + (if e = elem then 1 else 0)
;;
val compte : 'a list -> 'a -> int = <fun>
# compte [true; false; true; true; false] true;;
- : int = 3
# compte [true; false; true; true; false] false;;
- : int = 2
```

Remarque 1.17. On notera dans l'exemple précédent l'utilisation d'une expression avec if-then-else.

Remarque 1.18. On aurait pu vouloir indiquer directement un motif `elem::q`, cela ne fonctionne pas : les motifs ne doivent pas comporter de variables utilisées antérieurement ni deux fois la même variable.

Le filtrage peut être plus précis encore en indiquant des motifs plus spécifiques.

Exemple 1.37. La fonction suivante compte le nombre de fois qu'un élément recherché est présent deux fois consécutives :

```
# let rec couple liste elem = match liste with
  | [] -> 0
  | e1::e2::q when e1 = e2 && e1 = elem -> 1 + couple (e2::q) elem
  | e::q -> couple q elem
;;
val couple : 'a list -> 'a -> int = <fun>
# couple [1; 2; 2; 5; 2; 2; 2; -4; 0; 2] 2;;
- : int = 3
```

Remarque 1.19. On notera l'utilisation de `when` dès la description du motif à la place de l'utilisation d'un `if` ultérieurement.

1.6 Exceptions

Lorsqu'un comportement exceptionnel intervient (une erreur le plus souvent) il est signalé par une exception. On dit qu'une exception est *levée*.

Exemple 1.38.

```
# 1/0;;  
Exception: Division_by_zero.
```

Il est possible de *ratrapper* une exception avec la construction `try ... with .`.
À l'évaluation de l'expression `try expr with except -> val` on commence par évaluer l'expression `expr`. Si tout se passe bien, c'est cette valeur qui est renvoyée. Si l'exception `except` est levée, c'est la valeur `val` qui est renvoyée.

Exemple 1.39.

```
# try 1/0 with Division_by_zero -> 0;;  
- : int = 0
```

Remarque 1.20. On peut utiliser la même syntaxe que pour le filtrage de motifs dans le cas où plusieurs exceptions sont susceptibles d'être levées :

```
try expr with  
| except1 -> val1  
| except2 -> val2 ...
```

Il peut parfois être utile de lever soi-même une exception. on utilise pour cela le mot-clé `raise`.
Par ailleurs on peut déclarer de nouvelles exceptions selon la syntaxe `exception Mon_exception` voire `exception Mon_exception of type` s'il s'agit d'une exception avec argument.

Exemple 1.40.

```
# exception Droite_verticale;;  
exception Droite_verticale  
# let coef_dir a b = let (xa, ya) = a and (xb, yb) = b in  
    if xa = xb then raise Droite_verticale  
    else (yb -. ya) /. (xb -. xa) ;;  
val coef_dir : float * float -> float * float -> float = <fun>  
# coef_dir (0., 0.) (1., 4.);;  
- : float = 4.  
# coef_dir (1., 3.) (1., -2.);;  
Exception: Droite_verticale.
```

Remarque 1.21. Le nom de l'exception déclarée doit impérativement commencer par une majuscule.

Remarque 1.22. Une expression commençant par le mot-clé `raise` peut avoir un type différent selon le contexte, en cohérence avec le type de l'expression complète.

Dans le cas où l'exception levée n'est pas ratrapable et on souhaite indiquer un message d'erreur plus précis, on a recours à l'exception prédéfinie `Failure of string`. Celle-ci peut s'utiliser avec la syntaxe `failwith "message d'erreur"`. Ce qui correspond en fait à `raise (Failure "message d'erreur")`.

1.7 Types construits

Un nouveau type de données peut être déclaré à l'aide du mot-clé `type`.

1.7.1 Types énumérés (ou somme ou union)

Un type *énuméré* (ou encore type *somme* ou type *union*) est déclaré en indiquant l'ensemble (fini) des valeurs possibles de ce type, appelées *constructeurs*, séparées par un `|`.

Par exemple la déclaration `type saison = Printemps | Ete | Automne | Hiver` énumère les 4 valeurs possibles correspondant au type `saison`.

Remarque 1.23. Les constructeurs commencent obligatoirement par une majuscule, contrairement aux noms de variables qui commencent eux obligatoirement par une minuscule.

Remarque 1.24. L'ordre de déclaration des différents constructeurs induit une relation d'ordre sur le type énuméré. Ainsi on aura `Printemps < Ete < Automne < Hiver`.

Un constructeur peut éventuellement avoir des arguments, qu'on indique à l'aide du mot-clé `of`. Un type peut par ailleurs être construit récursivement. On peut avantageusement utiliser le filtrage de motif sur les types énumérés.

Exemple 1.41. On peut par définir un type `figure` comme ceci :

```
# type figure = None | Elem_Liste of (int * int) * figure;;
type figure = None | Elem_Liste of (int * int) * figure
# let rec nb_points fig = match fig with
    | None -> 0
    | Elem_Liste (point, sous_fig) -> 1 + nb_points sous_fig
;;
val nb_points : figure -> int = <fun>
# let triangle =
    Elem_Liste ( (3, 2), Elem_Liste ( (-1, 5), Elem_Liste ((0, 0), None) ));;
val triangle : figure =
    Elem_Liste ((3, 2), Elem_Liste ((-1, 5), Elem_Liste ((0, 0), None)))
# nb_points triangle;;
- : int = 3
```

1.7.2 Types enregistrements

Lorsque l'on souhaite définir des objets ayant plusieurs caractéristiques on peut déclarer leur type en renseignant entre accolades les différents *champs* possibles, en précisant à chaque fois le type associé.

La syntaxe est la suivante :

```
type mon_type_enr = { champs_1 : type_1 ; champs_2 : type_2 ; ... ; champs_n : type_n }
```

Pour déclarer une variable de type enregistrement on indiquera alors :

```
let var = { champs_1 = val_1 ; champs_2 = val_2 ; ... ; champs_n = val_n }
```

L'accès aux différentes valeurs de champs d'une variable s'effectuent ensuite en indiquant `var.champs` où `champs` est l'un des différents champs possibles.

Exemple 1.42.

```
# type rvb = { r : int ; v : int ; b : int };;
type rvb = { r : int; v : int; b : int; }
# let jaune = { r = 255; v = 255; b = 0 };;
val jaune : rvb = {r = 255; v = 255; b = 0}
# jaune.r;;
- : int = 255
```


Par défaut, les variables de type enregistrement sont immuables.

Si l'on souhaite donner un caractère mutable à certains champs, on l'indique au moment de la déclaration du type enregistrement à l'aide du mot-clé `mutable`. La valeur d'un champs peut alors être modifiée en indiquant `var.champs <- nouvelle_valeur`.

Exemple 1.43.

```
# type eleve = {nom : string; prenom : string; mutable age : int};;
type eleve = { nom : string; prenom : string; mutable age : int; }
# let alice = {nom = "Wonder"; prenom = "Alice"; age = 18};;
val alice : eleve = {nom = "Wonder"; prenom = "Alice"; age = 18}
# alice.age <- alice.age + 1;;
- : unit = ()
# alice;;
- : eleve = {nom = "Wonder"; prenom = "Alice"; age = 19}
# alice.nom <- "Dupond";;
Error: The record field nom is not mutable
```

Remarque 1.25. Les références vues précédemment sont en fait des enregistrements dont le champ `contents` est mutable.

Les types enregistrement peuvent aussi être définis récursivement.

Exemple 1.44.

```
# type personne = {prenom : string; enfants : personne list};;
type personne = { prenom : string; enfants : personne list; }
# let moi = {prenom = "Jade"; enfants = []};;
val moi : personne = {prenom = "Jade"; enfants = []}
# let mon_frere = {prenom = "Antoine"; enfants = []};;
val mon_frere : personne = {prenom = "Antoine"; enfants = []}
# let maman = {prenom = "Josephine"; enfants = [moi; mon_frere]};;
val maman : personne =
  {prenom = "Josephine";
   enfants =
     [{prenom = "Jade"; enfants = []}; {prenom = "Antoine"; enfants = []}]}
```

1.7.3 Types mutuellement récursifs

Pour déclarer des types *mutuellement récursifs*, c'est-à-dire qui sont construits l'un à partir de l'autre, on le fait conjointement avec le mot-clé `and`.

Exemple 1.45.

```
# type teinte = {nom : string; composition : couleur list} and couleur = Rouge | Vert
  | Bleu | Melange of teinte;;
type teinte = { nom : string; composition : couleur list; }
and couleur = Rouge | Vert | Bleu | Melange of teinte
# let purple = {nom = "violet"; composition = [Rouge; Bleu]};;
val purple : teinte = {nom = "violet"; composition = [Rouge; Bleu]}
# let peinture = Melange purple;;
val peinture : couleur =
  Melange {nom = "violet"; composition = [Rouge; Bleu]}
```

1.8 Ocaml sur sa propre machine

1.8.1 Installation de Ocaml

Sous Linux/macOS

On peut installer Ocaml via opam (gestionnaire de paquet spécialisé pour Ocaml) ou bien avec son gestionnaire de paquet habituel

Exemple 1.46. Sous Ubuntu :

```
sudo apt-get install opam
```

Exemple 1.47. Sous macOS avec Homebrew :

```
brew install opam
```

Opam doit être initialisé avec les commandes suivantes :

```
opam init -y  
eval $(opam env)
```

Sous windows

Suivre les instructions de la page <http://fdopen.github.io/opam-repository-mingw/installation/> (choisir l'installateur graphique version 32-bit ou 64-bit selon votre machine).

Cela va créer un environnement adapté Cygwin (suite de logiciels libres permettant de fournir un environnement type Unix) et y installer opam et Ocaml.

On initialise opam depuis le terminal Cygwin avec les commandes suivantes :

```
opam init -y  
eval $(opam env)
```

1.8.2 Boucle d'interaction (toplevel) et compilateur

Pour faciliter l'apprentissage de ce nouveau langage, nous avons jusqu'à présent utilisé un *interpréteur* Ocaml en mode interactif.

Pour utiliser une boucle d'interaction (ou *toplevel*) Ocaml depuis un terminal (éventuellement Cygwin) on saisit simplement la commande suivante :

```
$ ocaml
```

Ou bien, pour avoir la possibilité d'utiliser les flèches afin de récupérer la saisie précédente :

```
$ ledit ocaml
```

On obtient alors une information sur la version d'Ocaml utilisée ainsi qu'une invite # à la suite de laquelle il convient de saisir les expressions à évaluer :

```
$ ledit ocaml  
      OCaml version 4.02.3  
#
```

On quitte ce mode comme ceci :

```
# #quit;;  
$
```

Cependant, cette utilisation ne permet pas de sauvegarder ses lignes de code.

Pour exécuter un programme Ocaml enregistré sur sa machine on commence par l'enregistrer dans un fichier ayant une extension `.ml`.

Par exemple on peut enregistrer le programme ci-dessous dans un fichier nommé `hello.ml` :

```
let () = print_string "Hello world !\n"
```

Ce programme peut alors être exécuté dans un terminal avec la commande suivante :

```
$ ocaml hello.ml
```

Cette commande demande à l'interpréteur Ocaml d'exécuter le programme `hello.ml` ce qui a pour effet d'afficher la chaîne de caractères "Hello World!" comme prévu.

Remarque 1.26. Jusqu'à présent on utilisait systématiquement le double point-virgule `;` pour conclure la saisie des déclarations. Ce n'est plus utile si on ne passe pas par le mode interactif. Pour s'en passer il faut néanmoins s'assurer que notre programme correspond bien à une suite de déclarations.

Le programme suivant génère par exemple une erreur :

```
let a = 16
print_int a ; print_newline ()
```

En effet rien n'indique que la déclaration `let` est terminée à la fin de la première ligne, les expressions correspondant à l'affichage sont donc également incluses dans la déclaration.

On corrige cela aisément comme ceci :

```
let a = 16
let () = print_int a ; print_newline ()
```

Enfin, pour produire un fichier exécutable on utilise le **compilateur** Ocaml.

La commande qui convient est la suivante :

```
$ ocamlc hello.ml -o hello
```

Cette commande permet de compiler le code du fichier `hello.ml` et de produire du code exécutable par un ordinateur (en langage machine). On ajoute l'option `-o hello` pour préciser le nom que l'on souhaite donner au fichier exécutable généré. Ce dernier s'exécute ensuite avec la commande suivante :

```
$ ./hello
```

1.8.3 Utilisation de VSCode

VSCode est un éditeur de code multi-plateforme, open source, supportant une dizaine de langages de programmation. Il s'agit d'une distribution sous licence libre du logiciel VS Code (Visual Studio Code) de l'éditeur Microsoft.

Installer VSCode depuis <https://vscode.com/>

Sous Windows, cliquer sur « Download latest release », sélectionner la dernière version puis exécuter par exemple `VSCodeUserSetup-x64-1.71.2.22258.exe`

Il faut ensuite installer deux paquets permettant d'utiliser Ocaml depuis VSCode. On saisit pour cela dans un terminal (Cygwin pour Windows) la commande suivante :

```
opam install ocaml-lsp-server ocamlformat
```


Sous Windows, VSCode doit être lancé depuis l'environnement cygwin avec une commande du type (adapter selon l'emplacement de votre exécutable) :

```
"/cygdrive/c/Users/LCHALMAIN/AppData/Local/Programs/VSCode/VSCode.exe"
```

Le répertoire Unix `/cygdrive/c/Users/LCHALMAIN` correspond au chemin Windows `C:\Users\LCHALMAIN`. On peut également créer un raccourci sur le bureau puis dans les propriétés du raccourci éditer la cible en

ajoutant avant le chemin vers l'utilitaire `ocaml-env-win` :

```
C:\Ocaml64\usr\local\bin\ocaml-env-win.exe "C:\Users\LCHALMAIN\chemin\vers\VSCodium.exe"
```

Sur VSCode, ajouter les extensions Ocaml Platform (et éventuellement French Language Pack) en cliquant sur . On choisira également dans les paramètres (> Éditeur de texte > Suggestions) de ne pas accepter les suggestions après appui sur « Entrée ».

1.9 Modules

Ocaml dispose d'une bibliothèque fournissant un certain nombre de types et de fonctions. Cette bibliothèque est composée de **modules** correspondant à certains usages (manipuler des listes, des tableaux, etc.).

Un module est constitué de deux fichiers de même nom :

- Un fichier d'**implémentation** dont l'extension est `.ml`, contenant des déclarations de types et de valeurs.
- Un fichier d'**interface** dont l'extension est `.mli`, contenant les déclarations de types et les signatures des valeurs déclarées dans le fichier d'implémentation.

Les types et valeurs d'un module sont accessibles en utilisant le nom du module commençant par une majuscule et une notation pointée.

Exemple 1.48. La fonction `List.length` correspond à la fonction `length` déclarée dans le module `List` (dans le fichier `list.ml`).

Pour utiliser directement les éléments d'un module, sans cette notation, on peut éventuellement ouvrir ce module en indiquant `open module`. Attention dans ce cas à ce qu'il n'y ait pas de déclarations différentes pour un même identificateur dans des modules différents, il vaut donc mieux éviter cette pratique pour éviter toute ambiguïté.

Remarque 1.27. Sous Linux, on trouve la bibliothèque Ocaml à l'emplacement `/usr/lib/ocaml`.

Remarque 1.28. Il n'est pas attendu d'un élève de CPGE de connaître la bibliothèque Ocaml. Les fonctions disponibles dans les modules présentés ci-dessous peuvent être utilisées **à condition qu'il y en ait la mention explicite**.

1.9.1 Le module List

Les fonctions suivantes peuvent être utilisées après rappel :

Fonction	Signature	Description
<code>List.length</code>	<code>'a list -> int</code>	Renvoie le nombre d'éléments d'une liste
<code>List.mem</code>	<code>'a -> 'a list -> bool</code>	Teste si un élément est présent dans une liste ou non
<code>List.exists</code>	<code>('a -> bool) -> 'a list -> bool</code>	Teste si un prédicat est vérifié pour au moins l'un des éléments d'une liste
<code>List.for_all</code>	<code>('a -> bool) -> 'a list -> bool</code>	Teste si un prédicat est vérifié pour tous les éléments d'une liste
<code>List.filter</code>	<code>('a -> bool) -> 'a list -> 'a list</code>	Renvoie la liste des éléments d'une liste qui vérifient un certain prédicat
<code>List.map</code>	<code>('a -> 'b) -> 'a list -> 'b list</code>	Renvoie une liste correspondant aux éléments d'une liste auxquels une fonction a été appliquée
<code>List.iter</code>	<code>('a -> unit) -> 'a list -> unit</code>	Applique une procédure à chacun des éléments d'une liste

Remarque 1.29. La fonction `List.length` est explicitement au programme et donc toujours utilisable.

1.9.2 Le module Array

Les fonctions suivantes peuvent être utilisées après rappel :

Fonction	Signature	Description
<code>Array.length</code>	<code>'a array -> int</code>	Renvoie le nombre d'éléments d'un tableau
<code>Array.make</code>	<code>int -> 'a -> 'a array</code>	Initialise un tableau contenant à chaque fois la même valeur
<code>Array.copy</code>	<code>'a array -> 'a array</code>	Renvoie une copie superficielle d'un tableau
<code>Array.make_matrix</code>	<code>int -> int -> 'a -> 'a array array</code>	Initialise un tableau à 2 dimensions contenant à chaque fois la même valeur
<code>Array.init</code>	<code>int -> (int -> 'a) -> 'a array</code>	<code>Array.init n f</code> renvoie le tableau <code>[f 0; f 1; ...; f (n-1)]</code>
<code>Array.mem</code>	<code>'a -> 'a array -> bool</code>	Teste si un élément est présent dans un tableau ou non
<code>Array.exists</code>	<code>('a -> bool) -> 'a array -> bool</code>	Teste si un prédicat est vérifié pour au moins l'un des éléments d'un tableau
<code>Array.forall</code>	<code>('a -> bool) -> 'a array -> bool</code>	Teste si un prédicat est vérifié pour tous les éléments d'un tableau
<code>Array.map</code>	<code>('a -> 'b) -> 'a array -> 'b array</code>	Renvoie un tableau correspondant aux éléments d'un tableau auxquels une fonction a été appliquée
<code>Array.iter</code>	<code>('a -> unit) -> 'a array -> unit</code>	Applique une procédure à chacun des éléments d'un tableau

Remarque 1.30. Les fonctions `Array.length`, `Array.make` et `Array.copy` sont explicitement au programme et sont donc toujours utilisables.